0.0.1 S.N.Bernstein polynomials (1912):

$$B_n(f)(x) = \sum_{k=0}^{n} \binom{n}{k} x^k (1-x)^{n-k} f\left(\frac{k}{n}\right)$$

$$1 = ((1-x)+x)^n$$

Theorem 1 *If $f:[0,1] \to \mathbb{R}$ is a continuous function on $[0,1]$ than:*

$$\lim_{n\to\infty} \max_{x\in[0,1]} |B_n(f)(x) - f(x)| = 0.$$

$$B_n(f)(x) - f(x) = 0, x = 0, 1$$

Theorem 2 *f is decreasing (convex) on $[0,1] \Rightarrow B_n(f)$ is decreasing (convex) on $[0,1]$.*

Theorem 3 *If f is convex than:*

$$\begin{aligned} B_n(f)(x) &\geq f(x) \\ B_{n+1}(f)(x) &\leq B_n(f)(x) \end{aligned}$$

$$\begin{aligned} B_n(f)'(0) &= \frac{f(1/n) - f(0)}{1/n} \\ B_n(f)'(1) &= \frac{f(1-1/n) - f(1)}{-1/n} \end{aligned}$$

Theorem 4 *If $f^{(k)}$ is continuous on $[0,1]$ than*

$$\lim_{n\to\infty} \max_{x\in[0,1]} \left| B_n^{(k)}(f)(x) - f^{(k)}(x) \right| = 0.$$

Exercise 5 *Let $f(x) = -\sin(\pi x)$. Graphics of $f, B_n(f)$ for $n = 1, 6, 11, 16, 21, 26, ..., 331$. The same for $f(x) = -\sin(7\pi x)$*

Solution 6 *We propose the following commands:*

```
Bern[n_, f_, x_] :=
 Sum[Binomial[n, k]*x^k*(1 - x)^(n - k)*f[k/n], {k, 0, n}]
f[x_] := -Sin[Pi*x]
Animate[Plot[Evaluate[{f[x], Bern[n, f, x]}], {x, 0, 1},
 PlotStyle -> {RGBColor[1, 0, 0], RGBColor[0, 1, 0]}], {n, 1, 331,
 5}]
g[x_] := -Sin[7 Pi*x]
Animate[Plot[Evaluate[{g[x], Bern[n, g, x]}], {x, 0, 1},
 PlotStyle -> {RGBColor[1, 0, 0], RGBColor[0, 1, 0]}], {n, 1, 331,
 5}]
```

0.0.2 '1960 Pierre Bézier, Casteljeau:

Let $M_k\left(x_k, y_k\right)$, $k = 0, 1, .., n$, and $M = \left(M_k\right)_{k=0}^{n}$. BC is Bezier curve atached to the points M, given by:

$$BC\left(M\right)\left(t\right) = \sum_{k=0}^{n} \binom{n}{k} t^k \left(1 - t\right)^{n-k} \left(x_k, y_k\right)$$

Exercise 7 *Plot Bezier curve for points* $A\left(0, 0\right)$, $B\left(0, 1\right)$, $C\left(1, 1\right)$.

Solution 8 *We propose the following commands :*

```
Bez[A_, t_] :=
 Module[{n}, n = Dimensions[A][[1]] - 1;
 Sum[Binomial[n, k]*t^k*(1 - t)^(n - k)*A[[k + 1]], {k, 0, n}]]
A = {{0, 0}, {0, 1}, {1, 1}}
a = ListPlot[A, PlotStyle -> {RGBColor[1, 0, 1], PointSize[0.02]}]
b = ParametricPlot[Bez[A, t], {t, 0, 1}]
Show[a, b, AspectRatio -> 1]
```

0.0.3 The Popoviciu_Casteljeau algorithm:

Let points $M_k\left(x_k, y_k\right)$, $k = 0, 1, .., n$, and t fixed on $\left(0, 1\right)$.

We define the following triangular matrix:

for $i = 0$ $M_{k,i} = M_k, k = \overline{0, n}$

for $i > 0, i <= n$, $M_{k,i} = (1 - t)M_{k,i-1} + tM_{k+1,i-1}, k = \overline{0, n - i}$.

Than $M_{0,n}$ is the point of coordinates $\sum_{k=0}^{n} \binom{n}{k} t^k \left(1 - t\right)^{n-k} \left(x_k, y_k\right)$.

Exercise 9 *Plot the graph of the polygonal lines defined by Popoviciu-Casteljeau algorithm and the corresponding Bezier curve, with parameter* $s \in \left[0, t\right]$.

Solution 10 *:*

```
CC[m_, t_] :=
 Module[{n, M, a}, n = Dimensions[m][[1]] - 1;
 M = Table[0, {i, 0, n}, {j, 0, n - i}];
 a = Table[" ", {i, 0, n + 1}]; M[[1]] = m;
 a[[1]] = ListPlot[m, Joined -> True, DisplayFunction -> Identity];
 For[i = 1, i <= n, i++,
 For[j = 1, j <= n - i + 1, j++,
 M[[i + 1, j]] = t*M[[i, j + 1]] + (1 - t)*M[[i, j]]];
 a[[i + 1]] =
 ListPlot[M[[i + 1]], Joined -> True,
 PlotStyle -> RGBColor[i/n, 0, 1 - i/n],
 DisplayFunction -> Identity]];
 a[[n + 1]] =
 ListPlot[M[[n + 1]], PlotStyle -> PointSize[0.02],
```

```
 DisplayFunction -> Identity];
 a[[n + 2]] =
 ParametricPlot[
 Sum[Binomial[n, k]*s^k*(1 - s)^(n - k)*m[[k + 1]], {k, 0, n}], {s,
 0, t}, PlotStyle -> RGBColor[0, 1, 0],
 DisplayFunction -> Identity];
 Show[a, DisplayFunction -> $DisplayFunction,
 AspectRatio -> Automatic, PlotRange -> All]]
(*test*)
m = {{0, 0}, {0, 1}, {1, 1}, {1, 0}, {2, 2}};
CC[m, 1/2]
m = {{0, 0}, {0, 1}, {1, 1}, {1.5, -1}, {2, 1}, {3, 1}, {4, 0}};
Animate[CC[m, t], {t, 0.01, 1, 0.01}]
```